***RECORTES***

*Recorte geográfico (rurais do Norte)

generate rec_geo = 1 
replace rec_geo = 0 if (uf>=11 & uf<=16) & (v4728>=1 & v4728<=3)
label variable rec_geo "Recorte geográfico (tirando rurais do Norte)"

generate rec_idade = 0 
replace rec_idade = 1 if (v8005>=18 & v8005<=24)
label variable rec_idade "Recorte etário (selecionando 18-24 anos)"

generate rec_pos = 0 
replace rec_pos = 1 if (v0401==3)
label variable rec_pos "Recorte posição no domicílio (selecionando apenas filhos)"

generate rec_mod = 0 
replace rec_mod = 1 if (rec_geo==1 & rec_idade==1 & rec_pos==1)
label variable rec_mod "Recorte total para modelos"

***PESO***
generate peso = v4729
label variable peso "peso amostral"


***VARIÁVEIS INDEPENDENTES

*Versão PNAD
generate versao = 0
label define VERSAO 0 "PNAD antiga" 1 "PNADc"
label values versao VERSAO
label variable versao "VERSAO" 


*Ano
generate ano = .
replace ano = v0101
label variable ano "Ano da pesquisa" 

*Idade
generate idade = .
replace idade = v8005
label variable idade "Idade, em anos" 

*Sexo
generate sexo = .
replace sexo = 0 if v0302==2
replace sexo = 1 if v0302==4
label define SEXO 0 "Homens" 1 "Mulheres"
label values sexo SEXO
label variable sexo "Sexo" 

*Cor ou raça
generate cor = .
replace cor = 0 if (v0404==2 | v0404==6)
replace cor = 1 if (v0404==4 | v0404==8 | v0404==0)
label define COR 0 "Branco" 1 "Não-branco"
label values cor COR
label variable cor "Cor ou raça" 

*Região geográfica
generate regiao = .
replace regiao = 1 if (uf>=11 & uf<=17)
replace regiao = 2 if (uf>=21 & uf<=29)
replace regiao = 3 if (uf>=31 & uf<=35)
replace regiao = 4 if (uf>=41 & uf<=43)
replace regiao = 5 if (uf>=50 & uf<=53)
label define REGIAO 1 "Norte" 2 "Nordeste" 3 "Sudeste" 4 "Sul" 5 "Centro-Oeste"
label values regiao REGIAO
label variable regiao "Região geográfica" 

*Local
generate local = .
replace local = 0 if v4728>=1 & v4728<=3
replace local = 1 if v4728>=4 & v4728<=8
label define LOCAL 0 "Urbano" 1 "Rural"
label values local LOCAL
label variable local "Localização geográfica"

*Família monoparental
generate conj = 1 if v0401==2
by id_dom, sort : egen conj_agg = max(conj)

generate tipo_fam = .
replace tipo_fam = 0 if conj_agg==.
replace tipo_fam = 1 if conj_agg==1
label define FAMILIA 0 "Sem cônjuge" 1 "Com Cônjuge"
label values tipo_fam FAMILIA
label variable tipo_fam "Tipo de família"

*Escolaridade do chefe de domicílio
generate esc_pd8 = .
replace esc_pd8 = 15 if anoest>=15 & anoest<=17
replace esc_pd8 = anoest if anoest>=0 & anoest<=14

generate esc_chefe = .
replace esc_chefe = esc_pd8 if v0401==1
label variable esc_chefe "Anos de estudo (quando for chefe de domicílio)"

by id_dom, sort : egen esc_chefe_agg = max(esc_chefe)
label variable esc_chefe_agg "Anos de estudo do chefe do domicílio onde o entrevistado reside"

*Número de irmãos
generate filhos = 1 if v0401==3
by id_dom, sort : egen filhos_agg = sum(filhos)

generate filhos_agg2 = .
replace filhos_agg2 = filhos_agg-1 if (v0401==3)
label variable filhos_agg2 "Número de irmãos conviventes (somente para filhos)"

*CAT do responsável pelo domicílio
generate cat_chefe = .
replace cat_chefe = cat if v0401==1
label variable cat_chefe "CAT (quando for chefe de domicílio)"

by id_dom, sort : egen cat_chefe_agg = max(cat_chefe)
label variable cat_chefe_agg "CAT do chefe do domicílio onde o entrevistado reside"

label define CATC 1 "Grandes empregadores" 2 "Profissionais de nível superior" 3 "Pequenos empregadores" 4 "Ocupações médias" 5 "Trabalhadores do terciário especializado" 6 "Trabalhadores do secundário" 7 "Trabalhadores do terciário não-especializado" 8 "Trabalhadores agrícolas"
label values cat_chefe_agg CATC

*Renda domiciliar per capita
generate ind_mor = .
replace ind_mor = 1 if v0401>=1 & v0401<=5
label variable ind_mor "Indicador de morador (que não pensionistas, empregados domésticos, parentes dos empregados domésticos)"

by id_dom, sort : egen ind_mor_agg = sum(ind_mor)
label variable ind_mor_agg "Número de moradores (que não pensionistas, empregados domésticos, parentes dos empregados domésticos)"

generate renda_total_def_2022 = v4720def*1.80308493249454
label variable renda_total_def_2022 "Renda total deflacionada para 2022"

generate renda_total_ind = .
replace renda_total_ind = renda_total_def_2022 if (v0401>=1 & v0401<=5) & (v8005>=10)
label variable renda_total_ind "Renda total individual deflacionada para 2022 (que não pensionistas, empregados domésticos, parentes dos empregados domésticos)" 

by id_dom, sort : egen renda_total_agg = sum(renda_total_ind)
replace renda_total_agg = . if (v0401>=6 & v0401<=8)
label variable renda_total_agg "Renda total domiciliar (que não pensionistas, empregados domésticos, parentes dos empregados domésticos)"

generate renda_dom_pcta = renda_total_agg/ind_mor_agg
label variable renda_dom_pcta "Renda domiciliar per capita (que não pensionistas, empregados domésticos, parentes dos empregados domésticos)"



***VARIÁVEIS DEPENDENTES***

*Terminou o Ensino Médio?
generate ens_medio = . 
replace ens_medio = 0 if (curso_freq==1 | curso_freq==2 | curso_freq==3 | curso_freq==4 | curso_freq==6 | curso_freq==7) | (curso_nao_freq==1 | curso_nao_freq==2 | curso_nao_freq==4 | curso_nao_freq==8 | curso_nao_freq==9) | ((curso_nao_freq==3 | curso_nao_freq==5) & (v0611==3)) | ((curso_nao_freq==3 | curso_nao_freq==5) & (v0611==.) & (anoest>=0 & anoest<=10)) | ((curso_freq==. & curso_nao_freq==. & anoest==0))
replace ens_medio = 1 if (curso_freq==5 | curso_freq==9 | curso_freq==8) | (curso_nao_freq==6 | curso_nao_freq==7) | ((curso_nao_freq==3 | curso_nao_freq==5) & (v0611==1)) | ((curso_nao_freq==3 | curso_nao_freq==5) & (v0611==.) & (anoest>=11 & anoest<=17))
label define DIC 0 "Não" 1 "Sim"
label values ens_medio DIC
label variable ens_medio "Terminou o Ensino Médio?"


*Ingressou no Ensino Superior?
generate ens_sup = . 
replace ens_sup = 0 if (curso_freq==1 | curso_freq==2 | curso_freq==3 | curso_freq==4 | curso_freq==6 | curso_freq==7 | curso_freq==8) | (curso_nao_freq==1 | curso_nao_freq==2 | curso_nao_freq==3 | curso_nao_freq==4 | curso_nao_freq==5 | curso_nao_freq==8 | curso_nao_freq==9) | ((curso_freq==. & curso_nao_freq==. & anoest==0))
replace ens_sup = 1 if (curso_freq==5 | curso_freq==9) | (curso_nao_freq==6 | curso_nao_freq==7) 
label values ens_sup DIC
label variable ens_sup "Ingressou no Ensino Superior?"


*Frequenta nesse momento o Ensino Superior?
generate ens_sup_a = . 
replace ens_sup_a = 0 if (curso_freq==1 | curso_freq==2 | curso_freq==3 | curso_freq==4 | curso_freq==6 | curso_freq==7 | curso_freq==8) | (curso_nao_freq==1 | curso_nao_freq==2 | curso_nao_freq==3 | curso_nao_freq==4 | curso_nao_freq==5 | curso_nao_freq==6 | curso_nao_freq==7 | curso_nao_freq==8 | curso_nao_freq==9) | ((curso_freq==. & curso_nao_freq==. & anoest==0))
replace ens_sup_a = 1 if (curso_freq==5 | curso_freq==9)
label values ens_sup_a DIC
label variable ens_sup_a "Frequenta nesse momento o Ensino Superior?"


*Rede de ensino que estuda (APENAS PARA 2001 EM DIANTE)
generate rede_ens = .
replace rede_ens = 0 if (v6002==4) & (ens_sup_a==1)
replace rede_ens = 1 if (v6002==2) & (ens_sup_a==1)
label define REDE 0 "Privada" 1 "Pública"
label values rede_ens REDE
label variable rede_ens "Rede de ensino em que estuda (apenas para qum estava estudando no ensino superior)"




